/*
Rebecca A. Ferrer	
Proposal title: Conflict between curative and palliative goals among health care personnel: Elucidating psychological processes and resolving conflict through self-affirmation


HYPOTHESES

Stated-Hyp1: "Self-affirmation will reduce threat to medical identity; increase positive other-directed emotion and reduce the interpersonal empathy gap; and increase importance of palliative care goals." (p.6)

	Test-Hyp1: Threat to medical identity will be lower in self-affirmation condition vs. no affirmation condition.
	
	Test-Hyp2: Positive other-directed emption will be higher in self-affirmation condition vs. no affirmation condition.	

	Test-Hyp3: Empathy gap will be lower in self-affirmation condition vs. no affirmation condition.
	
	Test-Hyp4: The importance of palliative care goals will be higher in self-affirmation condition vs. no affirmation condition.

Stated-Hyp2: "The goal fusion induction will reduce threat to medical identity and increase the importance of palliative care goals." (p.6)

	Test-Hyp5: Threat to medical identity will be lower in goal fusion condition vs. no conflict condition.

	Test-Hyp6: The importance of palliative care goals will be higher in goal fusion condition vs. no conflict condition.
	
Stated-Hyp3: "Self-affirmation and goal fusion will interact, such that those in the self-affirmation condition will be most receptive to the goal fusion information, and will exhibit lowest identity threat and greatest importance of palliative care goals." (p.6)

	Test-Hyp7: Self-affirmation will moderate [increase] receptiveness to goal fusion in those exposed to goal fusion vs. conflict treatment.
	
	Test-Hyp8: Self-affirmation will moderate [decrease] identity threat in those exposed to goal fusion vs. conflict treatment. 

	Test-Hyp9: Self-affirmation will  moderate [increase] importance of palliative care goals in those exposed to goal fusion vs. conflict treatment. 

********************************************************************************
NOTES
- Proposal does not specify how the interpersonal empathy gap is measured. We 
are assuming it is the absolute difference between how someone would like to be 
treated themselves and how they would treat others.

- To measure receptiveness to goal fusion, we are creating a summated scale 
using seemingly relevant measures (was not specified in propoal). 	
*/

clear all
use "Ferrer825.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* Treatment group: 
	tab XTESS220
	recode XTESS220 (1/2=1 "1 self-affirmation") (3/4=0 "0 no affirmation"), gen(affirmation)
	recode XTESS220 (1 3=0 "0 conflict") (2 4=1 "1 fusion"), gen(fusion)
	tab affirmation
	tab fusion
	
* OUTCOME MEASURES

*threat to medical identity
	/*I feel less competent when I am faced with a patient who has a medical problem that I cannot solve.
	*/
	tab Q5, mis
	clonevar medid_threat=Q5
	replace medid_threat=. if medid_threat==-1

*positive other-directed emotion 
	/*
	1. How much do you currently feel nurturing?
	2. How much do you currently feel compassionate?
	*/
	
	replace Q4_1=. if Q4_1==-1
	replace Q4_2=. if Q4_2==-1
	gen otherdirected_emo=Q4_1+Q4_2
	tab otherdirected_emo
	

*interpersonal empathy gap
	/*
	1. The patients I see who are dealing with disease or illness feel it is important to seek:
	2. If you were a patient dealing with disease or illness, you would feel it is important to seek:
	*/
	replace Q7_1=. if Q7_1==-1
	replace Q7_2=. if Q7_2==-1	
	gen empathygap = abs(Q7_1-Q7_2)
	tab empathygap, mis


*importance of palliative care goals.
	* When thinking about the range of goals and priorities you may have for a patient's care, how much do you agree with the following statements? - 2 Palliative care is a strong priority:
	tab Q6_2
	replace Q6_2=. if Q6_2==-1
	clonevar pallcare_imp=Q6_2
 	
* receptiveness to goal fusion
	/* 1.	Does successfully achieving palliative goals for a patient have a helpful, harmful, or no effect at all on achieving curative goals for a patient?
	2.	Does successfully achieving curative goals for a patient have a helpful, harmful, or no effect at all on achieving palliative goals for a patient? 
	*/
	
	replace Q3_1=. if Q3_1==-1
	replace Q3_2=. if Q3_2==-1
	
	gen fusion_receptiveness= Q3_1+Q3_2
	tab fusion_receptiveness, mis
	
********************************************************************************

* ANALYSIS
	

*	Test-Hyp1: Threat to medical identity will be lower in self-affirmation condition vs. no affirmation condition.
	reg medid_threat i.affirmation
	// reject. 0.955
	tess 1.affirmation -, init(Ferrer825) bonf(4)
	
*	Test-Hyp2: Positive other-directed emption will be higher in self-affirmation condition vs. no affirmation condition.	
	reg otherdirected_emo i.affirmation
	// reject. 0.083
	tess 1.affirmation +, bonf(4)	
	
*	Test-Hyp3: Empathy gap will be lower in self-affirmation condition vs. no affirmation condition.
	reg empathygap i.affirmation
	// reject. 0.501
	tess 1.affirmation -, bonf(4)			
	
*	Test-Hyp4: The importance of palliative care goals will be higher in self-affirmation condition vs. no affirmation condition.
	reg pallcare_imp i.affirmation
	// reject. 0.333
	tess 1.affirmation +, bonf(4)		

*	Test-Hyp5: Threat to medical identity will be lower in goal fusion condition vs. no conflict condition.
	reg medid_threat i.fusion
	// reject. 0.759
	tess 1.fusion -, bonf(2)	
	
*	Test-Hyp6: The importance of palliative care goals will be higher in goal fusion condition vs. no conflict condition.
	reg pallcare_imp i.fusion
	// reject. 0.197	
	tess 1.fusion +, bonf(2)		

*	Test-Hyp7: Self-affirmation will moderate [increase] receptiveness to goal fusion in those exposed to goal fusion vs. conflict treatment.
	reg fusion_receptiveness i.fusion##i.affirmation
	// reject. 0.776
	tess 1.affirmation#1.fusion +, bonf(3)	
	
*	Test-Hyp8: Self-affirmation will moderate [decrease] identity threat in those exposed to goal fusion vs. conflict treatment. 
	reg medid_threat i.fusion##i.affirmation
	// reject. 0.104W
	tess 1.affirmation#1.fusion -, bonf(3)	
	
*	Test-Hyp9: Self-affirmation will  moderate [increase] importance of palliative care goals in those exposed to goal fusion vs. conflict treatment. 
	reg pallcare_imp i.fusion##i.affirmation
	// reject. 0.838
	tess 1.affirmation#1.fusion +, bonf(3)	
